CS225/226 SwitchingTheory - Aradhya Gupta (1901CS72)

Arduino based Mini Project

Touch Free Automatic Hand Sanitizer



Introduction

As we all know COVID-19 is spreading like a wild fire and all we hear is use a mask, practice social distancing and keep your hands clean. Hand sanitizer is the best way to get rid of all the harmful germs and viruses on our hands but what if a healthy person uses a contaminated dispenser which was previously used by an infected person. Then the same hand sanitizer becomes a cause for spreading the deadly virus.

This project is about creating a touch free automatic hand sanitizer dispenser to prevent the spread of coronavirus.

Aim

To implement an automatic hand sanitizer dispenser using an Arduino Uno board along with various other electronics components like DC pump and ultrasonic sensor, thus to familiarize with and understand their working principles and applications.

Motivation

The world is facing a tough time. COVID-19 has proved to be one of the deadliest pandemics humanity has ever faced. The need for touch free hand sanitizers has increased tenfold.

Components Used

1) Arduino Uno
2) HC-SR04 Ultrasonic Sensor
3) 12 volt DC Pump
4) JHD162A LCD display with I2C module
5) BC547B npn transistor
6) Buzzer
7) 12 volt DC adaptor
8) Red LED
9) Resistors (220 ohm)
10) Jumper wires and breadboard
11) Hand Sanitizer Container
12) Soldering iron

Circuit Diagram



Critical Analysis / Working principle

- Arduino Uno acts as the brains of the entire project. It is an open-source microcontroller board based on the Microchip ATmega328P microcontroller. The board is equipped with sets of digital and analog input/output (I/O) pins. It has 14 digital I/O pins, 6 analog I/O pins, and is programmable with the Arduino IDE.

- In this project digital pins 2 (echo pin), 3 (trigger pin), 5 (transistor pin), 6 (LED pin) and 8 (buzzer pin) have been used. Pin 2 acts as input pin while others act as output pins. Analog pins A4 (SDA) and A5 (SCL) have been used for LCD.
- HC-SR04 Ultrasonic sensor is used to detect the presence of a hand. It is capable of transmitting and receiving ultrasonic sound waves out into the air and calculates the distance from the sensor to the obstacle by making use of the time required for the sound wave to reach back the sensor. It has 4 pins, ground, VCC, trig and echo. The ground and the VCC pins of the module need to be connected to the ground and the 5 volts pins on the Arduino Board respectively and the trig and echo pins to any digital I/O pin on the Arduino Board.

  In order to generate the ultrasound trigger pin is set to a High State for 10 µs. This sends out an 8 cycle sonic burst which travels at the speed of sound and is received in the echo pin.

  The echo pin will output the time in microseconds the sound wave traveled. For example, if the object is 10 cm away from the sensor, and the speed of the sound is 340 m/s or 0.034 cm/µs the sound wave will need to travel about 294 u seconds. But what we get from the echo pin is double that number because the sound wave needs to travel forward and bounce backwards. So in order to get the distance in cm we multiply the received travel time value from the echo pin by 0.034 and divide it by 2.(refer source code)
- Once a hand is detected within 3 cm proximity, the transistor pin of the arduino is triggered to switch on the transistor thus providing a 12 volt power supply to the DC pump. Here the npn transistor amplifies the 5 volt input of the arduino board to 12 volt as required by the DC pump.
- After each use, the global variable ‘number’ is incremented to keep a count of the total number of uses and is shown on the LCD display. Once it reaches the threshold, the LED and buzzer pins are triggered and the circuit stops working. LCD displays the message to refill the container. The formula to calculate the threshold is

  $threshold = (capacity\ of\ container/amount\ dispensed\ in\ one\ use) - (a\ constant\ for\ spillage)$

  Here, threshold = (400ml/10ml) - 5 = 35 uses.

  In the code threshold is set to 3 just for the ease of demonstration.
- Power can be switched off and after refilling the container it can be turned on. The variable ‘number’ is reset to 0 and the circuit starts working again.

Source Code

```
//include libraries required for LCD display
#include <Wire.h>
#include <LiquidCrystal_I2C.h>

//define arduino pins to be used
#define echoPin 2
#define trigPin 3
#define transistorPin 5
#define ledPin 6
#define buzzerPin 8

//global variables
long duration;            // time taken by ultrasonic pulse to come back
int distance;             // distance of hand from ultrasonic sensor
long number = 0;          // number of times used
const int threshold = 3;  // maximum number of uses allowed before refill
                             (set to 3 for ease of demonstration)

// Set the LCD address to 0x27 for a 16 chars and 2 line display
LiquidCrystal_I2C lcd(0x27, 16, 2);

//set the modes(input or output) of used pins
void setup() {
  lcd.begin();
  pinMode(buzzerPin, OUTPUT);
  pinMode(trigPin, OUTPUT);
  pinMode(echoPin, INPUT);
  pinMode(transistorPin, OUTPUT);
  pinMode(ledPin, OUTPUT);
  Serial.begin(9600);
}
void loop() {
  lcd.clear();
  //calculate distance of hand from ultrasonic sensor
  digitalWrite(trigPin, LOW);
  delayMicroseconds(2);
  digitalWrite(trigPin, HIGH);
```

```
  delayMicroseconds(10);
  digitalWrite(trigPin, LOW);
  duration = pulseIn(echoPin, HIGH); //in micro secs

  distance = duration * 0.034 / 2; //speed of sound=340 m/sec
  Serial.print(" Dist: ");
  Serial.print(distance);
  Serial.print(" cm Cntr: ");
  Serial.print(number);
  Serial.print(" ");
  if(number<threshold){
    lcd.print("Number: ");
    lcd.print(number);
  }
  if(number>=threshold){
    digitalWrite(ledPin, HIGH); //turn the LED on
    lcd.clear();
    lcd.print("REFILL");
    tone(buzzerPin, 2000);      //buzzer emits sound of 2kHz frequency
    delay(2000);
    noTone(buzzerPin);
    Serial.println("LED on means 3 uses completed!");
  }
  else if (distance<=3){ //3 cm
    number++;
    digitalWrite(transistorPin, HIGH);  //turn the DC pump on
    Serial.print("Pump On--");
    delay(150);
    digitalWrite(transistorPin, LOW);
    Serial.println("--Pump OFF");
    delay(2000);
  }
  else{
    digitalWrite(transistorPin, LOW);
    Serial.println("--Pump OFF");
  }
  delay(50);
}
```

Applications

This project can be used at the entry of public places like malls, grocery stores, schools, etc. where a large number of people gather. Touch-free hand sanitizer prevents the spread of coronavirus by coming in contact with contaminated surfaces.

Future endeavours

The concept of threshold brings in some ambiguity and chance of error. Thus, a water sensor or even an ultrasonic sensor could be used to calculate the depth of liquid left in the container instead of hard-coding the threshold value initially. Then the user can use a container of any suitable capacity without having to change any value (like threshold).

References and Tools used

- Online Tutorial

  The project is inspired by this blog but has more features and components added to it such as LED, buzzer, LCD display and the feature to know when the dispenser gets empty using the concept of threshold.
- Alternate design using servo motor
- Soldering I2C module to LCD
- Tinkercad - online simulator for arduino
- Fritzing
- Library to integrate I2C module with LCD display

Aradhya Gupta

1901CS72

B. Tech. Sophomore Year

IIT Patna